

第 16 课 感动中国——数据文件

感动中国为我们诉说着一个个动人的故事，故事中的主人公给我们带来了太多太多的感动，他们是我们身边的榜样，他们是一些普通但又不平凡的人物。他们留给我们了一段段千古佳话。

在本课，我们将用 VB 来制作一个感动中国的小档案。如图 16.1 所示。




图 16.1 程序运行界面

Follow Me

1. 编写生成“感动中国人物”的顺序文件程序

分析：

- 搜集“感动中国人物”的资料，并将资料整理分为不同记录类型。
- 设计生成数据文件的窗体，以便顺序输入数据。
- 编写写入数据生成顺序文件的程序。

智多星

为了避免常用的数据在每次使用时重复输入，可以将常用的数据写入一个数据文件保存，待使用时从数据文件中读取。根据数据写入、读取规则的不同，数据文件一般分为顺序文件和随机文件。

步骤 1：设计写顺序文件的窗体，如图 16.2 所示。



图 16.2 写顺序文件窗体

在窗体中设计了五个文本框，用于输入每个感动人物的五项资料。当单击 Command1 按钮时将五项资料添加到顺序文件中。

步骤 3：写顺序文件的程序

结合写入数据的需要，设计关于命令按钮 Command1 的单击事件如下：

```
Private Sub Command1_Click()
    Rem 打开数据文件 sss.rtf，从标号为 1 的文件末尾准备写入数据
    Open App.Path & "\gdzg.rtf" For Append As #1
    Write #1, Text1.Text '将 Text1 中的数据写入标号为 1 的文件末尾
    Write #1, Text2.Text
    Write #1, Text3.Text
    Write #1, Text4.Text
    Write #1, Text5.Text
    Close #1 '关闭标号为 1 的数据文件
End Sub
```

智多星：

向一个数据文件写入数据或从一个数据文件读取数据时都需要首先用 Open 打开数据文件。在完成写入、读取数据的操作后还要用 Close 关闭该数据文件。

向顺序文件写入数据时可以从文件开头顺序写入，也可以从文件尾部顺序添加数据。是从开头写入数据，还是从尾部写入数据取决于打开方式的不同。

- 用 Append 方式打开的文件，数据从尾部开始写入。例如：
Open "<数据文件路径>" For Append As #<文件号>
- 用 Output 方式打开的文件。数据从开头开始写入。例如：
Open "<数据文件路径>" For Output As #<文件号>
从开头写入的数据将覆盖文件开头已有的数据。
- 向顺序文件中写入一个记录时可用：
Write #<文件号>, <变量>

步骤 3：搜集数据文件资料

根据“感动中国小档案”的程序设计要求，如图 16.1 在网上和其他媒体上，可以找到并整理成如下资料。

孟佩杰

女

2011 年感动中国

孝女当家

在贫困中，她任劳任怨，乐观开朗，用青春的朝气驱赶种种不幸；在艰难里，她无怨无悔，坚守清贫，让传统的孝道充满每个细节。虽然艰辛填满四千多个日子，可她的笑容依然灿烂如花。

张平宜

女

2011 年感动中国

希望之翼

蜀道难，蜀道难，台湾娘子上凉山。跨越海峡，跨越偏见，跨越怀疑，她抱起麻风村孤单的孩子，把无助的眼神柔化成对世界的希望。她看起来无比坚强，其实她的内心比谁都柔软。

.....

步骤 4: 运行程序，写入数据

16.3 将资料输入

运行程序如图 16.3，将搜集的资料输入或粘贴到对应的文本框中，单击标识为“增加顺序文件数据”的命令按钮，文本框中的数据会被写入数据文件 gdzg.rtf 中。

打开数据文件可以看到写入 gdzg.rtf 的数据格式如图 16.4 所示。

“孟佩杰”

“女”

“2011 年感动中国”

“孝女当家”

“在贫困中，她任劳任怨，乐观开朗，用青春的朝气驱赶种种不幸；在艰难里，她无怨无悔，坚守清贫，让传统的孝道充满每个细节。虽然艰辛填满四千多个日子，可她的笑容依然灿烂如花。”

“张平宜”

“女”

“2011 年感动中国”

“希望之翼”

“蜀道难，蜀道难，台湾娘子上凉山。跨越海峡，跨越偏见，跨越怀疑，她抱起麻风村孤单的孩子，把无助的眼神柔化成对世界的希望。她看起来无比坚强，其实她的内心比谁都柔”

图 16.4 写入数据文件的数据

2. 利用顺序文件的数据资料，完成“感动中国人物”小档案程序设计。

分析：

- 当用户按提示选择人物姓名后，相应的感动中国的数据信息就会显示。
- 选择姓名的控件使用组合框。
- 存放数据信息的文件用顺序文件


步骤 1：准备素材

将人物图片素材按照小档案人物顺序将图片更名为“1.jpg”…“3.jpg”……，并和“奖杯.jpg”的图片文件一起存放到“第 4 单元\16 感动中国\感动中国\”的文件夹中。

步骤 2：设计用户窗体界面，

在窗体上添加 3 个标签 (Label)，一个图片框 (PictureBox)，一个组合框 (ComboBox)，两个图像框 (Image)。按图 16.1 所示，摆放到合适位置，调整好大小，如图 16-5 所示。

如图 16.5。



16.5 为“感动中国小档案”设计的窗体界面

步骤 3: 设置对象属性

如表 16-1 对象属性设置表

表 16-1 对象属性设置表

对象	属性名	属性值	属性说明
Form1 (窗体)	名称	FrmStar	
	BorderStyle	1-Fixed Single	
	Caption	感动中国	
	StartUpPositon	2-屏幕中心	窗体在屏幕中心
	BackColor	黑色	
Label1 (标签框 1)	Caption	空	
	BackColor	白色	
	BorderStyle	1-Fixed Single	凹框
Label2 (标签框 2)	Caption	颁奖词	
	BackColor	黑色	
	ForeColor	白色	
Label3 (标签框 3)	Caption	感动中国人物	
	BackColor	黑色	
	ForeColor	黄色	
Picture1(图片框 1)	BackColor	白色	
Image1 (图像框 1)	BorderStyle	1-Fixed Single	
	BackColor	黑色	
	Stretch	True	
Image2 (图像框 2)	BackColor	黑色	
	Stretch	True	
Combo1 (组合框 1)	BackColor	白色	
	Text	选择感动中国人物	组合框显示的文字

步骤 4: 为组合框添加项目的程序代码。

在窗体的载入事件中初始化组合框, 为组合框添加项目。

```
Private Sub Form_Load()
    Combo1.AddItem "孟佩杰"
    Combo1.AddItem "张平宜"
    Combo1.AddItem "刘伟"
    Combo1.AddItem "胡忠与其妻谢晓君"
    .....
    Image2.Picture = LoadPicture(App.Path & "\奖杯.jpg") '加载奖杯图片
End Sub
```

步骤 5: 在组合框选择“感动中国人物”后执行的程序代码。

```
Private Sub Combo1_Click()
    n = Combo1.ListIndex + 1 '感动中国人物序号 n
    Open App.Path & "\gdzg.rtf" For Input As #1 '打开顺序文件读取数据
    For i = 1 To n
        Input #1, s1$ '将读取的姓名存放在变量 s1 中
```

```

Input #1, s2$           '将读取的性别存放在变量 s2 中
Input #1, s3$           '将读取的获奖时间性别存放在变量 s3 中
Input #1, s4$           '将读取的感动词存放在变量 s4 中
Input #1, s5$           '将读取的颁奖词存放在变量 s5 中
Next i
Close #1                 '关闭顺序文件
Picture1.Cls
Picture1.Print "姓名:" & s1$      '在图片框输出信息
Picture1.Print "性别:" & s2$
Picture1.Print "获奖时间:" & s3$
Picture1.Print "感动词:" & s4$
Label1.Caption = s5$      '在标签框输出信息
Image1.Picture = LoadPicture(n & ".jpg") '载入感动中国人物图片
End Sub

```

智多星

从顺序文件读取数据必须从文件头开始读取，“感动中国人物”每人的资料有 5 个记录，所以要获得第 2 个人的资料就要读取 10 条记录，若要获得第 n 个人的资料就要读取 n*5 条记录。为此在程序中用人物序号 n 来控制循环次数，每循环一次读取 5 个记录，即最后一次循环读出的记录是组合框中选择的第 n 个人物的 5 条记录资料。

- 打开顺序文件读取数据可用：
Open "<数据文件路径>" For Input As #<文件号>

打开数据文件。

- 从顺序文件中读取一个记录时可用：
Input #<文件号>, <变量>

步骤 6：保存文件

以“感动中国人物 1. frm”为窗体文件名存放在“第 4 单元\16 感动中国\感动中国\”的文件夹中。

智多星：

顺序文件是顺序存取的文件，顺序文件中的记录是按写入顺序，作为存放顺序和读取顺序的。即顺序文件中写在前面的记录，也会被存放在前面，读取记录时前面的记录也会最先被读出。若要读取顺序文件中的最后一个记录，必需从第一个记录开始依次读取到最后一个记录。

3. 编写、读、写“通讯录”的随机文件程序。

分析：

- 以随机文件的格式整理数据
- 分别设计写入随机文件和读取随机文件的窗体
- 分别设计读、写随机文件的程序

智多星：

与顺序文件不同随机文件各个记录的写入顺序和存储顺序以及读出的顺序都可能是不一样的，也就是说先写入的记录不一定存在前面，存在前面的记录也不一定先被读出。

随机文件就像一个表格，表中的每个记录都有一个记录号，通过记录号可以对相应的记

录进行读写。随机文件各数据记录长度是相同的，记录的各数据项的长度是固定的。

步骤 1：设计写入数据窗口

表 16-2 通讯录数据

编号 (No)	姓名(Name)	电话 (Phone)	邮箱 (Email)
1001	李向东	65834657	lixd2012@163.com
1002	高英	13659427631	gaoying@yaho.com
1003	王同强	40478543	wangtq@hotmail.com
1004	李焕	60557329	lihuan@126.com
.....

若要将表 16-2 中的数据写入随机文件中，不妨设计以下写入数据的窗体，如图 16.6 所示，窗体中主要有五个文本框 (TextBox) 和一个命令按钮 (CommandButton)。



图 16.6 向随机文件写入数据的窗体

步骤 2：定义数据记录类型的代码

在创建、打开一个随机文件前要先为随机文件自定义一个记录类型，在自定义记录类型时要对记录中每个字段定义数据类型及长度。为此可为存放通讯录的随机文件定义如下记录类型。

Rem 定义记录类型

```
Private Type address
    No As String * 4           ' 长度为 4 个字符的字符型数据
    Name As String * 8       ' 长度为 8 个字符的字符型数据
    Phone As String * 12    ' 长度为 12 个字符的字符型数据
    Email As String * 30    ' 长度为 30 个字符的字符型数据
End Type
```

步骤 3：写随机文件的代码

Rem 写随机文件

```
Private Sub Command1_Click()
    Dim t As address          ' 声明记录类型变量
    Open App.Path & "\add.rtf" For Random As #1 Len = Len(t) ' 打开随机文件
    i = Val(Text5.Text)      ' 获取记录号
    t.No = Text1.Text        ' 获取记录变量 t 中 No 字段的值
```

```

t.Name = Text2.Text      '获取记录变量 t 中 Name 字段的值
t.Phone = Text3.Text    '获取记录变量 t 中 Name 字段的值
t.Email = Text4.Text    '获取记录变量 t 中 Name 字段的值
Put #1, i, t            '按记录号写入随机文件
Close #1                '关闭随机文件
End Sub

```

在写入随机文件的代码中，首先声明了一个记录类型 address 的变量 t，变量 t 包含 4 个成员，它们是字段 No、字段 Name、字段 Phone 和字段 Email。各字段的数据通过文本框输入，获取的记录按记录号写入打开的随机文件。

通过以上程序的运行，可创建以下 add.rtf 的随机文件，如图 16.7 所示。

1001 李向东	65834657	lixd2012@153.com	1002 高英	13659427631
gaoying@yahoo.com.cn		1003 王同强	40478543	wangtq@hotmail.com
1004 李焕	60557329	lihuan@126.com		

图 16.7 生成的通讯录随机文件

智多星：

从随机文件读、写数据时需打开随机文件，并指定记录长度。对随机文件的读、写操作有如下规定。

- 用 Random 方式打开的文件，是随机文件。
格式：Open "<数据文件路径>" For Random As #<文件号> Len=<记录长度>
- 用 Close 关闭随机文件
格式：Close #<文件号>
- 用 Put 向随机文件中写入一个记录。
格式：Put #<文件号>, <记录号>, <变量>
- 用 Get 从随机文件中读取一个记录：
格式：Get #<文件号>, <记录号>, <变量>

步骤 4：保存通讯录写入随机文件

以“通讯录写随机文件.frm”为窗体文件名存放在“第 4 单元\16 感动中国\读写随机文件\”的文件夹中。

DIY：

参照写随机文件的程序，编写读取通讯录随机文件的程序，可参考以下窗体设计，如图 16.8 所示。




图 16.8 读取通讯录随机文件的参考窗体设计

4. 利用随机文件的数据资料，完成“感动中国人物”小档案程序设计。

步骤 1：利用“感动中国人物”资料创建资料数据的随机文件

提示：参照“通讯录写随机文件”程序。

步骤 2：保存“感动中国写入随机文件”程序

以“感动中国写随机文件.frm”为窗体文件名，存放在“第 4 单元\16 感动中国\感动中国\”的文件夹中。

步骤 3：设计程序，从“感动中国人物”的随机文件中读取数据，浏览资料，窗体设计如图 16.9。



图 16.9 从随机文件读取数据的“感动中国人物”浏览程序界面

步骤 4：保存“感动中国读取随机文件”程序

以“感动中国读随机文件.frm”为窗体文件名，存放在“第 4 单元\16 感动中国\感动中国\”的文件夹中。

知识着陆

数据文件的打开与关闭

- 以 Output 方式打开的顺序文件。数据从开头开始写入。
格式：Open "<数据文件路径>" For Output As #<文件号>
- 以 Append 方式打开的顺序文件，数据从尾部开始写入。
格式：Open "<数据文件路径>" For Append As #<文件号>
- 以 Input 方式打开的顺序文件，数据从开头开始读取。
格式：Open "<数据文件路径>" For Input As #<文件号>
- 以 Random 方式打开的文件是随机文件。
格式：Open "<数据文件路径>" For Random As #<文件号> Len=<记录长度>
- 用 Close 关闭数据文件。
格式：Close #<文件号>

数据记录的写入与读取

- 用 Write 向顺序文件中写入一个记录。
格式：Write #<文件号>, <变量>

- 用 Input 从顺序文件中读取一个记录。
格式： Input #<文件号>, <变量>
- 用 Put 向随机文件中写入一个记录。
格式： Put #<文件号>, <记录号>, <变量>
- 用 Get 从随机文件中读取一个记录。
格式： Get #<文件号>, <记录号>, <变量>

视野拓展

文件扩展名是操作系统用来标志文件格式的一种机制。通常来说，一个扩展名是跟在主文件名后面的，由一个分隔符分隔。在一个像“readme.txt”的文件名中，readme 是主文件名，txt 为扩展名，表示这个文件被认为是一个纯文本文件，常见文件的扩展名如表 16-3 所示。

表 16-3 常见文件扩展名

文件扩展名	文件类型
doc	Word 文档，用微软的 word 等软件打开
wps	Wps 文档，用金山公司的 wps 软件打开
xls	Excel 电子表格，用微软的 excel 软件打开
ppt	Powerpoint 演示文稿，用微软的 powerpoint 等软件打开
txt	纯文本，用记事本、写字板、word 等都可以打开
rar	WinRAR 压缩文件，用 WinRAR 等打开
htm	网页文件，用浏览器、Microsoft Office FrontPage 打开
pdf	用 pdf 阅读器打开、用 pdf 编辑器编辑
exe	可执行文件
jpg	图形文件
bmp	位图文件
swf	flash 影片

创作天地

1. 建立家庭成员顺序文件资料档案，编写浏览家庭成员档案资料的程序。
2. 建立本小组 10 名成员的体检随机文件资料档案，编写浏览体检资料的程序。
3. 读取体检资料，统计 10 名组员的平均身高。
4. 读取体检资料，找出 10 名组员中体重最重的组员。
5. 读取体检资料，根据 10 名组员的身高，由高到低的排列记录资料的顺序。